Pensieve header: Weave knots computations.

Following Roland’s MatTheta3.nb.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\Theta"];
<< Theta.m
```

Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.

In[∘]:=
```
(*Example calculation*)
ThetaTrefoil = Θ[{{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0, 0}}]
```

Out[∘]=

$$\left\{\frac{1 - T + T^2}{T}, \frac{1 - T_1 + T_1^2 - T_2 - T_1^3 T_2 + T_2^2 + T_1^4 T_2^2 - T_1 T_2^3 - T_1^4 T_2^3 + T_1^2 T_2^4 - T_1^3 T_2^4 + T_1^4 T_2^4}{T_1^2 T_2^2}\right\}$$

In[∘]:=
```
PolyPlot[ThetaTrefoil]
```

Out[∘]=



## Some more complicated knots

A mat-shaped knot based on an m by n rectangular grid. The integers m,n should be even and the crossings can be chosen arbitrarily.
Probably all knots can be brought into this form.

In[ ]:=


m,n even
n
3n
5n
(m-1)
n
5=n-1
4
3
y
2
1
mn+m
0
mn
2n
4n
x
0
1
2
3
4
5
6
7=m-1
mn+y m+n-x
x n + y
mn+y m+n-x
mn+y m+x
x n + y
mn+y m+x

Out[ ]=


m,n even
n
3n
5n
(m-1)
n
5=n-1
4
3
y
2
1
mn+m
0
mn
2n
4n
x
0
1
2
3
4
5
6
7=m-1
mn+y m+n-x
x n + y
mn+y m+n-x
mn+y m+x
x n + y
mn+y m+x

In[◦]:=
```
(*m,n, even, by default all xings positive.  The
 list negxs flips the sign of those crossings.*)
Weave[m_, n_, negxs_ : {}] := Module[{pd}, pd = (Flatten@Table[
        Switch[
         {EvenQ[x], EvenQ[y]},
         {True, True},
         X[m n + y m + m - x - 1, n x + y + 1, m n + y m + m - x, n x + y],
         {True, False},
         X[n x + y, m n + y m + x + 1, n x + y + 1, m n + y m + x],
         {False, True},
         X[n x + n - y - 1, m n + y m + m - x, n x + n - y, m n + y m + m - x - 1],
         {False, False},
         X[m n + y m + x, n x + n - y, m n + y m + x + 1, n x + n - y - 1]
        ]
        , {x, 0, m - 1}, {y, 0, n - 1}]
    ) /. {X[a_, b_, c_, d_] :> X[a + 1, b + 1, c + 1, d + 1]};
  Do[pd[[i]] = (pd[[i]] /. {X[a_, b_, c_, d_] :> X[d, a, b, c]}), {i, negxs}];
  PD @@ pd
 ]
(*Randomly assign signs to all crossings*)
RandWeave[m_, n_] := Weave[m, n, RandomSample[Range[m n], RandomInteger[{0, m n}]]]
```

In[◦]:=
```
DrawXing[x_, y_, cut_] :=
 If[cut === "v", {Line[{{x, y - 1/2}, {x, y - 1/4}}], Line[{{x, y + 1/4}, {x, y + 1/2}}],
   Line[{{x - 1/2, y}, {x + 1/2, y}}]}, {Line[{{x, y - 1/2}, {x, y + 1/2}}],
   Line[{{x - 1/2, y}, {x - 1/4, y}}], Line[{{x + 1/4, y}, {x + 1/2, y}}]}]
```

In[ ]:=
```
DrawWeave[m_, n_, negxs_ : {}] := Graphics[{Thickness[0.1 / Max[m, n]],
   Table[
    Switch[
     {EvenQ[x], EvenQ[y]},
     {True, True},
     DrawXing[x, y, If[MemberQ[negxs, y m + x + 1], "v", "h"]],
     {True, False},
     DrawXing[x, y, If[MemberQ[negxs, y m + x + 1], "h", "v"]],
     {False, True},
     DrawXing[x, y, If[MemberQ[negxs, y m + x + 1], "h", "v"]],
     {False, False},
     DrawXing[x, y, If[MemberQ[negxs, y m + x + 1], "v", "h"]]
    ]
    , {x, 0, m - 1}, {y, 0, n - 1}],
   Table[{Arrowheads[{{0, .7}}],
     Arrow[BezierCurve[{{x, -1/2}, {x, -1}, {x + 1, -1}, {x + 1, -1/2}}]]}, {x, 1, m - 3, 2}],
   Table[{Arrowheads[{{0, .7}}], Arrow[
      BezierCurve[{{x, n - 1/2}, {x, n}, {x + 1, n}, {x + 1, n - 1/2}}]]}, {x, 0, m - 2, 2}],
   Table[{Arrowheads[{{0, .7}}],
     Arrow[BezierCurve[{{-1/2, y}, {-1, y}, {-1, y + 1}, {-1/2, y + 1}}]]}, {y, 0, n - 2, 2}],
   Table[{Arrowheads[{{0, .7}}], Arrow[
      BezierCurve[{{m + -1/2, y}, {m, y}, {m, y + 1}, {m + -1/2, y + 1}}]]}, {y, 1, n - 3, 2}],
   BezierCurve[{{m - 1, -1/2}, {m - 1, -1}, {m, 0}, {m - 1/2, 0}}],
   {Arrowheads[{{0, 1}}], Arrow[{{m - 1/2, n - 1}, {m, n - 1}}]},
   {Arrowheads[{{0, 0.5}}], Arrow[{{0, -1}, {0, -1/2}}]}

  }]
DrawRandWeave[m_, n_] := DrawWeave[m, n, RandomSample[Range[m n], RandomInteger[m n]]]
```

In[ ]:= **DrawWeave**[8, 6]

Out[ ]=



In[ ]:=
```
DrawWeave[8, 6, {1, 6}]
DrawRandWeave[20, 20]
```

Out[ ]=



Out[ ]=



In[∘]:= 
```
DrawRandWeave[50, 50]
```
Out[∘]=



In[∘]:= 
```
DrawRandWeave[30, 30]
```
Out[∘]=



```
RandWeavePair[m_, n_] := Module[{nxs = RandomSample[Range[m n], RandomInteger[m n]]},
  {DrawWeave[m, n, nxs], PolyPlot@Θ@Weave[m, n, nxs]}]
```

In[ ]:= 
```
GraphicsGrid[Table[RandWeavePair[4, 2], {j, 7}] // Transpose]
```

Out[ ]=



In[ ]:= 
```
GraphicsGrid[Table[RandWeavePair[6, 8], {j, 7}] // Transpose]
```

Out[ ]=



In[ ]:= 
```
(*The knots Weave[m,n] are both alternating and also positive,
and thus fibered. This seems to make their theta look especially simple. The
 sides of the hexagon (i.e. the coefficient of the maximal power of T2) are
 in this case conjectured to be always equal to half the signature times
 the Alexander polynomial in T1. Below we print the first few square ones.*)
```

In[ ]:= 
```
SquareWeaves =
  GraphicsGrid@Transpose@Table[{DrawWeave[n, n], PolyPlot@Θ@Weave[n, n]}, {n, 4, 12, 2}]
```

Out[ ]=



In[◦]:=
```
Export["figs/SquareWeaves.pdf", SquareWeaves]
```
Out[◦]=
```
figs/SquareWeaves.pdf
```

In[◦]:=
```
Select[Range[100], Random[] < 0.5 &]
```
Out[◦]=
```
{2, 3, 4, 5, 7, 8, 16, 19, 21, 23, 25, 31, 33, 36, 38, 39, 40, 42, 44, 48, 49, 53,
 54, 61, 62, 65, 66, 72, 75, 77, 78, 80, 84, 86, 87, 88, 89, 90, 96, 97, 98, 100}
```

In[◦]:=
```
RandomWeave = With[{m = 12, n = 12},
  Module[{nxs = Select[Range[m n], Random[] < 0.5 &]},
   GraphicsRow[{DrawWeave[m, n, nxs], PolyPlot@Θ@Weave[m, n, nxs]}]]
  ]
```
Out[◦]=



In[◦]:=
```
Export["figs/RandomWeave.pdf", RandomWeave]
```
Out[◦]=
```
figs/RandomWeave.pdf
```

In[∘]:=
```
GraphicsRow[Table[
   ImageCompose[
    PolyPlot[Θ[Weave[n, n]], ImageSize → 480],
    Show[DrawWeave[n, n], ImageSize → 240],
    {Right, Bottom}, {Right, Bottom}],
   {n, 2, 10, 2}]]
```

Out[∘]=